

Академия Управления при Президенте Республики Беларусь
Институт государственного управления

Кафедра информационных технологий

Специальность "Государственное управление и экономика"

КУРСОВАЯ РАБОТА

по курсу "Экономико-математические методы и модели"

на тему:

*"Сетевое планирование и управление
в анализе экономических ситуаций"*

Выполнил:
студент 4 курса, 3 группы,
заочной формы обучения,
специальность 123456

Иванов Иван
Иванович

Преподаватель:

Петров Петр
Петрович

Минск 2011

СОДЕРЖАНИЕ

Введение.....	3
1. Основные понятия сетевого планирования и управления.....	6
2. Правила построения сетевых графиков	8
3. Расчет временных параметров сетевого графика	12
4. Постановка задачи.....	15
5. Описание метода решения	17
6. Программа расчета на языке VBA.....	19
7. Результаты расчета сетевого графика	22
8. Экономическая интерпретация результатов	24
Заключение	25
Список литературы	26

Введение

В современных условиях при реализации многих проектов выполняются тысячи взаимосвязанных работ-операций. Сети оказались естественным и удобным средством для описания и анализа сложных проектов при экономико-математическом моделировании.

Сетевые модели сложных комплексов работ были разработаны начали использоваться совсем недавно, в 50-х годах прошлого столетия.

Сетевая модель была применена в США при создании баллистических ракет «Поларис», предназначенных для оснащения атомных подводных лодок американского военно-морского флота. В сложном комплексе работ при этом участвовало свыше 6000 фирм, работы выполнялись на территории 48 штатов Америки, а сетевой график включал в себя более 10 000 событий. Сейчас в США сетевые методы планирования и управления получили широкое распространение. Первая и наиболее распространенная система планирования и управления в США носит название системы «ПЕРТ» (PERT, Program Evaluation and Review Technique – оценка программ и способ проверки), вторая - СРМ (Critical Path Method - метод критического пути). Преимущества, полученные в результате их применения, оказались настолько значительными, что в США ни одна строительная фирма не ведет строительство без использования системы «ПЕРТ» или ее разновидностей. Ни одна фирма не получает правительственного контракта на выполнение новых заказов, если она не освоила систему «ПЕРТ». Американские экономисты подсчитали, что сети позволяют на треть сократить продолжительность работ.

В нашей стране разработаны несколько иные системы планирования и управления, коротко называемые «СПУ». В основе этих систем тоже лежат сетевые графики. Применяются и другие методы организационного управления, но системы «СПУ» получили наибольшее распространение. Системы «СПУ» были успешно применены, например, при сооружении ТЭЦ в Лисичанске, Буштырской тепловой электростанции, Челябинского блюминга-автомата

«1300», при ремонте мартеновской печи завода «Серп и молот», при реконструкции доменной печи в «Запорожстали», при строительстве метромоста через Днепр в Киеве, комплекса жилых и общественных зданий на проспекте Калинина в Москве и т. д.

В настоящее время все большее число строек, предприятий, научно - исследовательских институтов и проектных организаций страны переключаются на планирование и оперативное управление крупными комплексами работ с помощью систем «СПУ».

Обозримость сетевого графика или его частей значительно облегчает восприятие существа всей системы, взаимосвязей всех работ, упрощает весь последующий процесс по руководству системой при ее реализации. Дело, конечно, не в самом сетевом графике, а в той организационной системе, которая осуществляется с его помощью.

Сети небольшого объема с успехом могут обрабатываться вручную или с использованием калькуляторов. Сети с большим числом событий обрабатываются, конечно, с помощью ЭВМ. Если система СПУ охватывает тысячи событий, то специалисты составляют «частные» графики. ЭВМ «сшивают» их в огромную единую сеть, которую никто и никогда не видит. В таких случаях всю тяжесть расчетов и сопоставлений человек перекладывает на ЭВМ. Она становится единственной хранительницей информации о всем сетевом графике. Человек, вложив в машину данные о каждой конкретной работе, может получить от машины все нужные ему сведения.

Конечно, системы «СПУ» не идеальны, они пока не дают возможности вести управление при одновременном учете всех параметров (и времени, и стоимости, и ресурсов, и технико-экономических показателей), тем не менее их методы являются весьма эффективными.

Методы «СПУ» отвечают потребностям тех, кто имеет дело с выполнением проектов или крупных комплексов работ. Это один из методов исследования операций, но он не требует никаких предварительных специальных знаний. Все понятия, с которыми приходится сталкиваться при его использовании, просты и

воспринимаются интуитивно. Специалисты считают системы «СПУ» крупнейшим за последние годы достижением научной организации труда и утверждают, что за системами «СПУ» большое будущее.

В этой курсовой работе будут изложены и продемонстрированы на примере основные принципы построения и расчета сетевых графиков, лежащих в основе систем сетевого планирования и управления.

1. Основные понятия сетевого планирования и управления

При планировании сложных комплексов взаимосвязанных и взаимообусловленных работ и управления ходом их выполнения наиболее эффективны методы сетевого планирования и управления (СПУ). Доступность и простота этих методов позволяют широко использовать их в практической работе.

В основе методов СПУ лежит графическое представление проекта (комплекса работ для достижения поставленной цели) в виде сетевого графика. Сетевой график можно рассматривать как совокупность G некоторого количества точек $E_1, E_2, \dots$ и соответственно между ними установленных связей $\bar{e}_1, \bar{e}_2, \dots$. Объект G называется *графом*, точки $E_1, E_2, \dots$ - его *вершинами*, связи между ними $\bar{e}_1, \bar{e}_2, \dots$ - *дугами*. Ориентация дуг, т. е. указание "начала" и "конца" каждой из них, делает граф *ориентированным (орграфом)*. Граф $G = (E, g)$ считается заданным, если заданы все его вершины и дуги. Будем отождествлять вершины орграфа с событиями, а дуги - работами. События и работы - основные понятия в СПУ.

Работа - это любые действия, трудовые процессы, сопровождающиеся затратами ресурсов или времени и приводящие к определенным результатам. На сетевых графиках работы изображают отрезками прямых линий с указанием направления (т. е. стрелками). Рядом со стрелкой указываются числовые характеристики: время выполнения работы, расход ресурса, количество исполнителей и т. д. Под работами подразумеваются не только реальные хозяйственные или технологические процессы, требующие затрат времени и ресурсов для их осуществления, но и процессы, потребляющие только время. Например, естественная сушка материалов, затвердение бетона и т. п. Также принято считать работами и те процессы, которые не требуют ни затрат времени, ни ресурсов. Это так называемые *зависимости*, или *фиктивные работы*. Они показывают, что одна работа не может совершаться раньше другой. На сетевых графиках фиктивные работы обычно изображают пунктирными стрелками.

Событие обозначает факт окончания всех работ в него входящих или начала работ из него выходящих. Событие не имеет протяженности во времени. На сетевом графике события изображаются геометрическими фигурами (круги, квадраты). Чаще кругами с указанием номера или шифра события. В каждое событие может входить и выходить из него несколько работ, а каждая работа ограничена двумя событиями. Событие выражает логическую связь между работами, заключающуюся в том, что работы, входящие в данное событие, непосредственно предшествуют работам, выходящим из него; ни одна выходящая из данного события работа не может начинаться до окончания всех работ, входящих в это событие.

Событие, с которого начинается выполнение проекта, является *исходным*, оно не имеет предшествующих работ. Событие, которое констатирует факт завершения проекта, называется *завершающим*, оно не имеет последующих работ. Все прочие события являются *промежуточными*.

Предполагается, что события не имеют продолжительности и наступают как бы мгновенно.

Моментом свершения события считается момент окончания выполнения всех входящих в это событие операций. Пока не выполнены все входящие в событие операции, не может свершиться само событие, а следовательно, не может быть начата ни одна из следующих непосредственно за ним операций.

Различают три вида операций:

1. *действительная операция* ($\longrightarrow$) - процесс, требующий затрат времени и ресурсов (разработка проекта, подвоз материалов, выполнение монтажных работ и т.д.);

2. *операция ожидания* ($-\cdots-\rightarrow$) - процесс, требующий только затрат времени (затверждение бетона, естественная усушка штукатурки перед началом малярных работ, рост растений и т.д.);

3. *фиктивная операция* ($-----\rightarrow$), или логическая зависимость, отражает технологическую либо ресурсную зависимость при выполнении некоторых операций.

2. Правила построения сетевых графиков

Построение сетевого графика начинается с составления списка операций (работ), подлежащих выполнению (см. табл. 1). Последовательность операций в списке произвольная. Порядок нумерации операций осуществляется в соответствии с последовательностью их в списке. Операции, включенные в список, характеризуются определенной продолжительностью, которая устанавливается на основе действующих нормативов или по аналогии с ранее выполнявшимися операциями. Такие временные оценки называются *детерминированными*. Если же нормативные данные временных оценок операций отсутствуют, то определяются *вероятностные оценки*.

После составления списка операций приступают непосредственно к процедуре построения. При построении сетевого графика необходимо соблюдать определенные правила.

1. В сети не должно быть событий (кроме исходного), в которые не входит ни одна дуга.

2. Не должно быть событий (кроме завершающего), из которых не выходит ни одной дуги.

3. Сеть не должна содержать контуров, так как ни одна операция не может предшествовать сама себе.

4. Любая пара событий сетевого графика может быть соединена не более чем одной дугой. Если изобразить одновременно (параллельно) выполняемые три различные операции b, c, d с общими начальным и конечным событиями (рис. 1), то возникает путаница из-за того, что различные операции имеют один и тот же шифр (2, 5).




Рис. 1.

В этом случае рекомендуется ввести дополнительные события и соединить их с последующими фиктивными операциями (рис. 2).




Рис. 2.

5. Номер начального события любой операции должен быть меньше номера ее конечного события.

6. Если какие-либо операции могут быть начаты до полного окончания непосредственно предшествующей им операции, то последнюю целесообразно представить как ряд последовательно выполняемых операций, завершающихся определенными событиями.

Например, если операции с и d могут быть начаты до полного окончания операции b, то операцию b рекомендуется разбить на элементарные операции b_1 , b_2 , b_3 и представить выполнение всех операций в виде графика, изображенного на рис. 3.




Рис. 3.

7. Для отражения технологической или ресурсной зависимости при выполнении операций применяют фиктивные операции.

Предположим, что операция с может выполняться после завершения операций а и b, а операция d — только после завершения операции b. Эта зависимость представлена на рис. 4.




Рис. 4.

Из рисунка видно, что операция с следует за операцией а и фиктивной операцией (2,3). В свою очередь операция (2,3) следует за операцией b. Тогда в силу транзитивности выполнение операции b предшествует выполнению операции с.

Построенный с соблюдением этих правил график является сетевой моделью выполнения проекта. При этом сначала обычно составляются частные сетевые графики, охватывающие работы по отдельным, имеющим самостоятельное значение частям общего комплекса работ, а затем путем "сшивания" получается комплексный (сводный) график, охватывающий всю совокупность работ, подлежащих выполнению.

На построенном графике работы можно идентифицировать номерами начального и конечного событий (i, j), причем, события имеют упорядоченную нумерацию, т.е. для любой работы (i, j) выполняется неравенство $i < j$. Но обычно, после того, как сеть построена, для удобства работы с ней, нужно перенумеровать события. Для этого можно воспользоваться графическим спосо-

бом упорядочения вершин графа по рангам (методом вычеркивания дуг). Графический способ состоит в следующем:

1. Исходную вершину (в которую не входит ни одна дуга) отнесем к рангу 0 и присвоим ей номер 1.

2. Вычеркнем все дуги, выходящие из вершины (1) и отнесем события, оказавшиеся без входящих дуг, к первому рангу. Этим событиям присвоим в произвольном порядке номера 2, 3, ... k_1 .

3. Вычеркнув все дуги, выходящие из вершины предыдущего ранга i , отнесем вершины, оказавшиеся без входящих дуг, к следующему $i + 1$ -му рангу. Присвоим им номера $k_i + 1, \dots k_{i+1}$. Этот шаг повторяем до тех пор пока все вершины не будут пронумерованы.

3. Расчет временных параметров сетевого графика

Для управления ходом выполнения комплекса операций, представленного сетевой моделью, необходимо располагать количественными параметрами элементов сети. К таким параметрам относятся: продолжительность выполнения всего комплекса операций, сроки выполнения отдельных операций и их резервы времени. Важнейшим параметром сетевого графика является также критический путь. *Путь* на сетевом графике называется последовательность дуг, в которой начало каждой последующей дуги совпадает с концом предыдущей. Путь можно записать в виде последовательности вершин, через которые он проходит (например, 1 - 2 - 3 - 5 - 7). Различают следующие виды путей: полный, предшествующий событию, следующий за событием.

Путь сетевого графика называется *полным*, если его начальная вершина совпадает с исходным событием, а конечная - с завершающим.

Предшествующий событию путь - это путь от исходного события до данного.

Следующий за событием путь есть путь от данного события до завершающего.

Критическим называется полный путь, имеющий наибольшую продолжительность во времени. Операции и события, принадлежащие критическому пути, называются соответственно *критическими операциями* и *критическими событиями*. Суммарная продолжительность операций, принадлежащих критическому пути, равна *критическому времени* $t_{кр}$ выполнения комплекса операций в целом. На графике критический путь, как правило, выделяется жирной или двойной линией.

Расчет основных временных параметров производится по соответствующим формулам. Существует множество методов расчета, в данной курсовой работе используем математический метод динамического программирования.

Ранний срок $t_p(j)$ свершения события j - это самый ранний момент, к которому завершаются все работы, предшествующие этому событию

$$t_p(j) = \max_{i,j} (t_p(i) + t_{ij}), \quad (1)$$

где максимум определяется по множеству работ, заканчивающихся j - м событием; $t_p(i)$ - ранний срок свершения начального события работы (i, j) ; t_{ij} - продолжительность работы (i, j) . Полагаем $t_p(1) = 0$.

Поздний срок $t_n(i)$ свершения события i - такой предельный момент, после которого остается ровно столько времени, сколько необходимо для выполнения всех работ, следующих за этим событием

$$t_n(i) = \min_{i,j} (t_n(j) - t_{ij}), \quad (2)$$

где минимум определяется по множеству работ, начинающихся i - м событием; $t_n(j)$ - поздний срок свершения конечного события работы (i, j) .

Резерв времени $R(i)$ события i показывает, на какой предельно допустимый срок может задержаться свершение события i без нарушения срока наступления завершающего события

$$R(i) = t_n(i) - t_p(i). \quad (3)$$

Критический путь - это наиболее протяженный во времени полный путь, его продолжительность и определяет критический срок $t_{кр}$. Для критических событий ранние и поздние сроки совпадают, резерв времени равен нулю.

Ожидаемые и предельные сроки свершения событий находятся в диалектическом единстве со сроками начала и окончания операций: ранний срок начала выполнения операции (i, j) равен ожидаемому сроку свершения i -го события; поздний срок окончания операции совпадает с поздним сроком свершения ее конечного события; поздний срок начала выполнения операции равен разно-

сти между предельным сроком свершения ее конечного события и продолжительностью; ранний срок окончания операции равен сумме ожидаемого срока свершения ее начального события и продолжительности.

Сроки выполнения операций находятся в границах, определяемых указанными выше параметрами. Следовательно, операции, как и события, могут иметь некоторый резерв времени. Различают четыре разновидности резервов времени операций: полный, свободный, частный первого вида и частный второго вида.

Полный резерв времени операции $R_n(i, j)$ показывает, насколько можно сдвинуть начало выполнения операции или увеличить ее продолжительность, не изменяя ожидаемого срока свершения начального события, при условии, что конечное для данной операции событие свершится не позднее своего предельного срока. Полный резерв времени вычисляется по формуле

$$R_n(i, j) = t_n(j) - t_p(i) - t_{i,j} \quad (4)$$

Свободный резерв времени операции $R_c(i, j)$ показывает, насколько можно увеличить продолжительность или отсрочить начало выполнения операции (i, j) при условии, что начальное и конечное ее события свершаются в ожидаемое время

$$R_c(i, j) = t_p(j) - t_p(i) - t_{i,j} \quad (5)$$

Свободный резерв времени присущ только данной работе, и его использование никак не повлияет на выполнение последующих работ. Только отдельные работы проекта обладают свободным резервом времени.

4. Постановка задачи

Рассмотрим прикладную задачу построения и анализа укрупненного сетевого графика комплекса операций по реконструкции цеха. Список операций представлен в таблице 1.

Таблица 1.

Операция	Шифр операции	Наименование операции	Опирается на операции	Продолжительность, дн.
a ₁	(1,2)	Подготовительные работы	-	5
a ₂	(1,3)	Демонтаж старого оборудования	-	3
a ₃	(2,6)	Ремонтные строительно-монтажные работы	a ₁	30
a ₄	(3,4)	Подготовка фундамента под оборудование	a ₁ , a ₂	16
a ₅	(2,4)	Подготовка к монтажу нового оборудования	a ₁	10
a ₆	(2,5)	Электротехнические работы	a ₁	12
a ₇	(4,5)	Монтаж нового оборудования	a ₄ , a ₅	8
a ₈	(5,7)	Подключение оборудования к электросети	a ₆ , a ₇	2
a ₉	(7,8)	Наладка и технологические испытания оборудования	a ₈	6
a ₁₀	(6,8)	Отделочные работы	a ₃ , a ₆ , a ₇	8
a ₁₁	(8,9)	Приемка цеха в эксплуатацию	a ₉ , a ₁₀	1

Сетевой график комплекса операций изображен на рисунке 5.




Рис. 5. Сетевой график комплекса работ.

Все операции графика, за исключением фиктивных операций (2,3) и (5,6), являются действительными. Числа в скобках, приписанные дугам, означают продолжительность выполнения соответствующих операций. Операции a_1 и a_2 не опираются ни на какие операции, следовательно, на графике они изображаются дугами, выходящими из исходного события (1), означающего момент начала выполнения комплекса операций. Операции a_3 , a_5 и a_6 опираются на операцию a_1 , поэтому на графике дуги a_3 , a_5 и a_6 следуют непосредственно за дугой a_1 . Событие (2) означает момент окончания операции a_1 и начала операций, представленных дугами, выходящими из этого события. Операция a_4 опирается на операции a_1 и a_2 . Графически это условие отражено посредством последовательного изображения операций (1,3) и (3,4) и введения фиктивной операции (2,3). Событие (3) инцидентно операциям (1,3) и (2,3), следовательно, моментом свершения события (3) является такой момент, к которому будут выполнены все входящие в это событие операции и может быть начата операция, отраженная дугой, выходящей из него. Аналогично с учетом технологии выполнения изображены на графике остальные операции. Завершающее событие (9) означает момент окончания выполнения всего комплекса операций по реконструкции цеха. Шифры операций (см. таблицу 1) состоят из номеров начального и конечного событий и заносятся в список после составления графика.

5. Описание метода решения

Для вычисления временных параметров сетевого графика по уравнениям раздела 4 используем табличный процессор Excel.

Исходные данные для вычисления следующие:

- n – число событий;
- m – число операций (работ);
- (i, j) – обозначения работ;
- $t_{i,j}$ – продолжительности работ.

Представим эти данные в виде электронной таблицы, показанной на рисунке 6.

	A	B	C	D	E	F	G	H	I
1			Таблица исходных данных						
2									
3		Номер	Начальное	Конечное	Продолжи-	Расчет			
4		операции	событие	событие	тельность				
5		1	1	2	5				
6		2	1	3	3				
7		3	2	3	0				
8		4	2	4	10				
9		5	2	5	12				
10		6	2	6	30				
11		7	3	4	16				
12		8	4	5	8				
13		9	5	6	0				
14		10	5	7	2				
15		11	6	8	8				
16		12	7	8	6				
17		13	8	9	1				
18									
19									

Рис. 6. Окно Excel с таблицей исходных данных для расчета.

Поместим на рабочий лист электронной таблицы кнопку Расчет. При нажатии на эту кнопку программа осуществит необходимый расчет сетевого графика работ и выведет на рабочий лист таблицу временных параметров событий

и таблицу временных параметров работ. За таблицами выводим значения критического срока и критический путь.

Алгоритм расчета временных параметров сетевого графика должен обеспечивать выполнение следующих задач:

- ввод исходных данных;
- по исходным данным вычисление ранних сроков свершения событий по формуле (1);
- вычисление поздних сроков свершения событий по формуле (2);
- определение резервов времени событий по формуле (3);
- определение критического пути;
- вычисление полного резерва времени работ по формуле (4);
- вычисление свободного резерва времени работ по формуле (5).

Список используемых переменных имеет вид:

$t[i, j] = t_{ij}$ - продолжительности работ;

$tp[i] = t_p(i)$ - ранний срок свершения событий;

$tn[i] = t_n(i)$ - поздний срок свершения событий;

st - критический путь.

6. Программа расчета на языке VBA

```
Private Sub CommandButton1_Click()

Dim t(1 To 20, 1 To 20) As Integer
Dim tp(1 To 20) As Integer
Dim tn(1 To 20) As Integer
Dim flag As Boolean
Dim st As String, st1 As String

' Подсчет числа работ m по числу строк исходной таблицы
m = 0
Cells(5, 2).Select
While Not IsEmpty(ActiveCell.Value)
    m = m + 1
    ActiveCell.Offset(1, 0).Select
Wend

' Поиск числа событий n как максимума среди конечных событий
n = -1
For i = 1 To m
    If Cells(i + 4, 4).Value > n Then n = Cells(i + 4, 4).Value
Next i

' Задаем начальные значения матрицы продолжительности работ
For i = 1 To n
    For j = 1 To n
        t(i, j) = -1
    Next j
Next i

' Вводим данные о работах из исходной таблицы
For i = 1 To m
    t(Cells(i + 4, 3).Value, Cells(i + 4, 4).Value) = Cells(i + 4, 5).Value
Next i

' Расчет ранних сроков событий
tp(1) = 0
For i = 2 To n
    tp(i) = -1
Next i
Do
    For i = 2 To n
        If tp(i) < 0 Then

            ' Определяем, найдены ли все ранние сроки событий,
            ' предшествующих i-му событию
            flag = True
            For j = 1 To n - 1
                If (t(j, i) >= 0) And (tp(j) < 0) Then flag = False
            Next j
            If flag Then
                ' находим ранний срок tp(i)
                For j = 1 To n - 1
                    If (t(j, i) >= 0) And (tp(i) < tp(j) + t(j, i)) Then tp(i) = tp(j) + t(j, i)
                Next j
            End If
        End If
    Next i
Loop Until tp(n) > 0

' Расчет поздних сроков событий
```

```

MaxInt = 32766
tn(n) = tp(n)
For i = 1 To n - 1
    tn(i) = MaxInt
Next i
Do
    For i = 1 To n - 1
        If tn(i) = MaxInt Then

            ' Определяем, найдены ли все поздние сроки событий,
            ' следующих за i-м событием
            flag = True
            For j = 2 To n
                If (t(i, j) >= 0) And (tn(j) = MaxInt) Then flag = False
            Next j
            If flag Then
                ' находим поздний срок tn[i]
                For j = 2 To n
                    If (t(i,j)>=0) And (tn(i)>tn(j)-t(i, j)) Then tn(i) = tn(j) - t(i, j)
                Next j
            End If
        End If
    Next i
Loop Until tn(1) < MaxInt

' Создаем заголовок таблицы "Временные параметры событий"
Cells(7 + m, 3).Value = "Временные параметры событий"
Cells(7 + m, 3).Font.Size = 14
Cells(7 + m, 3).HorizontalAlignment = xlLeft
Cells(9 + m, 2).Value = "Номер"
Cells(9 + m, 3).Value = "Ранний"
Cells(9 + m, 4).Value = "Поздний"
Cells(9 + m, 5).Value = "Резерв"
Cells(10 + m, 2).Value = "события"
Cells(10 + m, 3).Value = "срок"
Cells(10 + m, 4).Value = "срок"
Cells(10 + m, 5).Value = "времени"

' Заполняем таблицу "Временные параметры событий"
For i = 1 To n
    Cells(10 + m + i, 2).Value = i
    Cells(10 + m + i, 3).Value = tp(i)
    Cells(10 + m + i, 4).Value = tn(i)
    Cells(10 + m + i, 5).Value = tn(i) - tp(i)
Next i

' Разлинейваем таблицу "Временные параметры событий"
Range(Cells(9 + m, 2), Cells(10 + m + n, 5)).Select
Selection.Borders.Color = RGB(0, 0, 0)
Selection.Borders(xlInsideHorizontal).LineStyle = xlNone

Range(Cells(9 + m, 2), Cells(10 + m, 5)).Select
Selection.Borders.Color = RGB(0, 0, 0)
Selection.Borders(xlInsideHorizontal).LineStyle = xlNone

' Создаем заголовок таблицы "Временные параметры работ"
Cells(13 + m + n, 3).Value = "Временные параметры работ"
Cells(13 + m + n, 3).Font.Size = 14
Cells(13 + m + n, 3).HorizontalAlignment = xlLeft
Cells(15 + m + n, 2).Value = "Номер"
Cells(15 + m + n, 3).Value = "Начальное"
Cells(15 + m + n, 4).Value = "Конечное"
Cells(15 + m + n, 5).Value = "Продолжи-"

```

```

Cells(15 + m + n, 6).Value = "Полный"
Cells(15 + m + n, 7).Value = "Свободный"
Cells(16 + m + n, 2).Value = "операции"
Cells(16 + m + n, 3).Value = "событие"
Cells(16 + m + n, 4).Value = "событие"
Cells(16 + m + n, 5).Value = "тельность"
Cells(16 + m + n, 6).Value = "резерв"
Cells(16 + m + n, 7).Value = "резерв"

' Заполняем таблицу "Временные параметры работ"
k = 0
For i = 1 To n
    For j = 1 To n
        If t(i, j) >= 0 Then
            k = k + 1
            Cells(16 + m + n + k, 2).Value = k
            Cells(16 + m + n + k, 3).Value = i
            Cells(16 + m + n + k, 4).Value = j
            Cells(16 + m + n + k, 5).Value = t(i, j)
            Cells(16 + m + n + k, 6).Value = tn(j) - tp(i) - t(i, j)
            Cells(16 + m + n + k, 7).Value = tp(j) - tp(i) - t(i, j)
        End If
    Next j
Next i

' Разлинейваем таблицу "Временные параметры событий"
Range(Cells(15 + m + n, 2), Cells(16 + 2 * m + n, 7)).Select
Selection.Borders.Color = RGB(0, 0, 0)
Selection.Borders(xlInsideHorizontal).LineStyle = xlNone

Range(Cells(15 + m + n, 2), Cells(16 + m + n, 7)).Select
Selection.Borders.Color = RGB(0, 0, 0)
Selection.Borders(xlInsideHorizontal).LineStyle = xlNone

' Находим критический путь
st = "1-"
i = 1
Do
    flag = True
    For j = 2 To n
        If (t(i, j) >= 0) And (tn(j) = tp(j)) And flag Then

            flag = False
            i = j
            st1 = Str(j)
            If j < n Then st = st + st1 + "-" Else st = st + st1
        End If
    Next j
Loop Until i = n

Cells(18 + 2 * m + n, 2).Value = "Критический срок ="
Cells(18 + 2 * m + n, 2).HorizontalAlignment = xlLeft
Cells(18 + 2 * m + n, 4).Value = tp(n)
Cells(19 + 2 * m + n, 2).Value = "Критический путь ="
Cells(19 + 2 * m + n, 2).HorizontalAlignment = xlLeft
Cells(19 + 2 * m + n, 4).Value = st


Range("A1").Select

End Sub

```

7. Результаты расчета сетевого графика

При запуске расчета сетевого графика комплекса операций по реконструкции цеха кнопкой Расчет получаем показанные на рисунках 7 и 8 созданные таблицы.



The screenshot shows a Microsoft Excel window titled "Microsoft Excel - Расчет сетевого графика.xls". The active sheet is "Лист1". The table displayed is titled "Временные параметры событий" (Event Time Parameters) and is located in the range A22:E32. The table has 5 columns: "Номер события" (Event Number), "Ранний срок" (Early Date), "Поздний срок" (Late Date), and "Резерв времени" (Time Reserve). The data is as follows:

Номер события	Ранний срок	Поздний срок	Резерв времени
1	0	0	0
2	5	5	0
3	5	11	6
4	21	27	6
5	29	35	6
6	35	35	0
7	31	37	6
8	43	43	0
9	44	44	0

Рис. 7. Таблица временных параметров событий.

Microsoft Excel - Расчет сетевого графика.xls									
Файл Правка Вид Вставка Формат Сервис Данные Окно Справка 100% 12 Ж К Ч Объект...									
A16	fx								
	A	B	C	D	E	F	G	H	I
35			Временные параметры работ						
36									
37		Номер	Начальное	Конечное	Продолжи-	Полный	Свободный		
38		операции	событие	событие	тельность	резерв	резерв		
39		1	1	2	5	0	0		
40		2	1	3	3	8	2		
41		3	2	3	0	6	0		
42		4	2	4	10	12	6		
43		5	2	5	12	18	12		
44		6	2	6	30	0	0		
45		7	3	4	16	6	0		
46		8	4	5	8	6	0		
47		9	5	6	0	6	6		
48		10	5	7	2	6	0		
49		11	6	8	8	0	0		
50		12	7	8	6	6	6		
51		13	8	9	1	0	0		
52									
53		Критический срок =		44					
54		Критический путь =		1- 2- 6- 8- 9					
55									
56									
Лист1 Лист2 Готово NUM									

Рис. 8. Таблица временных параметров работ.

8. Экономическая интерпретация результатов

В результате расчета получен критический путь 1-2-6-8-9. На критическом пути лежат операции подготовительных работ (1,2), ремонтных строительно-монтажных работ (2,6), отделочных работ (6,8) и приемки цеха в эксплуатацию (8,9). Их суммарная продолжительность и составляет критический срок: $5+30+8+1=44$ дня. Это минимальное время, необходимое для выполнения всего комплекса операций по реконструкции цеха.

Зная критические операции и временные параметры всех других операций, можно оптимизировать проект по времени с целью сокращения критического срока, то есть выполнение всех работ меньше, чем за 44 дня. Для этого нужно с помощью дополнительных вложений или перераспределения трудовых и технических ресурсов уменьшить время выполнения операций, лежащих на критическом пути.

Например, подготовительные работы лежат на критическом пути и занимают 5 дней. В это же время выполняется демонтаж старого оборудования, для чего требуется 3 дня. Если часть рабочих перевести с демонтажа старого оборудования на подготовительные работы, то эти две операции можно выполнить, к примеру, за 4 дня. Тем самым время выполнения всех работ сократится на один день и составит 43 дня.

Заключение

Экономико-математическое моделирование – исключительно важный эффективный метод анализа экономических явлений, планирования экономических систем и управления ими. По сравнению со многими отраслями естественных наук экономические науки представляют особенно плодотворную область использования моделирования.

Математическое моделирование является кибернетическим методом познания исследователем экономического процесса с использованием своеобразного инструмента – экономико-математической модели (ЭММ). В широком смысле под ЭММ понимается концентрированное выражение в математической форме существенных взаимосвязей и закономерностей функционирования экономического процесса (экономической системы).

В настоящее время разработаны основные типы математических моделей как народнохозяйственного и отраслевого, так и регионального и внутризаводского (внутрихозяйственного) планирования.

По своему назначению ЭММ делят на теоретико-аналитические и прикладные. Теоретико-аналитические ЭММ используются при исследовании самых общих закономерностей в экономических процессах, прикладные - при решении конкретных задач прогнозирования, управления и экономического анализа. По способу отражения связей и соотношений различают ЭММ структурные и функциональные. Структурные модели отражают внутреннюю организацию системы и используются при описании систем со сложной структурой организации. Важнейшее назначение функциональных моделей - познание сущности объекта. Построение функциональной модели предполагает поиск оператора, связывающего вход и выход системы. Внутренняя структура системы в функциональных моделях не отражается.

Следует отметить, что развитие экономических исследований с применением математических методов сопровождается появлением новых типов ЭММ, новых признаков их классификации и интеграции.

Список литературы

1. Гранберг А. Г. Динамические модели народного хозяйства. Москва, "Экономика", 1985.
2. Гринберг А. С., Гурина Т. Н. Экономико-математические модели и методы. Минск, АУ при КМ РБ, 1994.
3. Математическая экономика на персональном компьютере. Под. ред. М. Кубонива. "Финансы и статистика", Москва, 1991.
4. Терехов Л. Л. Экономико-математические методы. - М., 1968.
5. Экономико-математические методы и модели. Под ред. А. В. Кузнецова, Минск, БГЭУ, 1999.